
	Reason for change:
	In the current TTCN implementation of function : f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common

1) v_DrbIdList is initialized at the beginning of the function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common  and the initialized value is getting used in Steps 17a.1 and 18.1. 
It is proposed to update v_DrbIdList value before making use of it in steps 17a.1 and 18.1

2) At step 18.1, SRB DRB release is expected to happen on Source Scell instead of Source Pcell

3) In case of 8.1.4.1.7.1, during the test steps 7- 8 , 
NR Cell 3  NR Cell 23 handover happens with NR Cell 2 gets released as Scell . and scell (NR cell 2 ) was born as scell inactive ( only SRB0 was configured ).
It is proposed to add extra checks before doing f_NR_SS_SRBs_DRBs_Release at Step 18.1

4) In case of 8.1.4.1.7.1,  during the test steps 4-5, NR Cell 1  NR Cell 3 handover happens with NR Cell 2 added as Scell ; when target Pcell is same as Source Scell , SRB DRB reset on Source Pcell is happening twice in the function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common 
· 1st time SRB DRB RESET :: inside f_NR_ReconfigurePCellToNormalCell which is called from 4.1,5th steps of f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common
· 2nd  time SRB DRB RESET :: as part of 17.1 step of f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common )  this can be improvised by blocking the redundant RB reset and release inside f_NR_ReconfigurePCellToNormalCell based on the condition
· This way we can also make sure RB reset is done on source Cell only after the PDCP HO competition in align with other HO cases

	
	

	Summary of change:
	1) v_DrbIdList is updated every time before it gets used in the function calls in Steps 17a.1 and 18.1
2) At step 18.1, SRB DRB release is expected to happen on Source Scell 
3) RB Count check is added before doing f_NR_SS_SRBs_DRBs_Release at Step 18.1
4) SRB DRB reset in the function f_NR_ReconfigurePCellToNormalCell is guarded with the condition so that it avoids the RB RESET on Source Pcell happening twice when source Scell is becoming target Pcell

1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc106808477]Corrections required for NR testcases 8.1.4.1.7.x and 8.1.4.1.8.x
1.1 [bookmark: _Toc106808478]Correction to the function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common
	Function name
	f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common


	Reason for change
	In the current TTCN implementation of function : f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common

1) v_DrbIdList is initialized at the beginning of the function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common  and the initialized value is getting used in Steps 17a.1 and 18.1. 
It is proposed to update v_DrbIdList value before making use of it in steps 17a.1 and 18.1

2) At step 18.1, SRB DRB release is expected to happen on Source Scell instead of Source Pcell

3) In case of 8.1.4.1.7.1, during the test steps 7- 8 , 
NR Cell 3  NR Cell 23 handover happens with NR Cell 2 gets released as Scell . and scell (NR cell 2 ) was born as scell inactive ( only SRB0 was configured ).
It is proposed to add extra checks before doing f_NR_SS_SRBs_DRBs_Release at Step 18.1

4) In case of 8.1.4.1.7.1,  during the test steps 4-5, NR Cell 1  NR Cell 3 handover happens with NR Cell 2 added as Scell ; when target Pcell is same as Source Scell , SRB DRB reset on Source Pcell is happening twice in the function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common 
· 1st time SRB DRB RESET inside f_NR_ReconfigurePCellToNormalCell which is called from 4.1,5th steps of f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common (  with activation time )
· 2nd  time SRB DRB RESET as part of 17.1 step of f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common )  this can be improvised by blocking the redundant RB reset and release inside f_NR_ReconfigurePCellToNormalCell based on the condition ( with immediate activation )
· This way we can also make sure RB reset is done on source Cell only after the PDCP HO competition in align with other HO cases 

	Summary of change
	1) v_DrbIdList is updated every time before it gets used in the function calls in Steps 17a.1 and 18.1
2) At step 18.1, SRB DRB release is expected to happen on Source Scell 
3) RB Count check is added before doing f_NR_SS_SRBs_DRBs_Release at Step 18.1
4) SRB DRB reset in the function f_NR_ReconfigurePCellToNormalCell is guarded with the condition so that it avoids the RB RESET on Source Pcell happening twice when source Scell is becoming target Pcell

	TTCN module
	NR5GC_Handover.ttcn

	MCC160 Comment
	



Before Change:
	function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common(NR_CellId_Type                               p_SourcePCellId,
                                                          NR_CellId_Type                               p_SourceSCellId,
                                                          NR_CellId_Type                               p_TargetPCellId,
                                                          template(omit) NR_CellId_Type                p_TargetSCellId,
                                                          template(omit) CellGroupConfig               p_SCGConfig_SCell,
                                                          RNTI_Value                                   p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                          boolean                                      p_RBConfig_KeyChange := true,
                                                          template (value) DL_DCCH_Message             p_RRCReconfiguration,
                                                          template (value) NR_RachProcedureConfig_Type p_RachProcedureConfig,
                                                          template (value) NR_DRBInfoList_Type         p_DrbInfoList,
                                                          template (value) UL_GrantConfig_Type         p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_OnSR,
                                                          template (value) UL_GrantConfig_Type         p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_OnSR,
                                                          template (present) UL_DCCH_Message           p_RRCConnReconfigComplete := cr_38508_RRCReconfigurationComplete,
                                                          NextHopChainingCount                         p_NCC := tsc_NR_38508_NextHopChainingCount,
                                                          template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit
                                                      ) runs on NR_BASE_PTC
  { //@sic R5s201249: added p_SCGConfig_SCell R5-213516: added MacBearerRouting sic@
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing1;
    var SubFrameTiming_Type v_Timing2;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList := f_NR5GC_GetRadioBearerList(p_DrbInfoList);
    var IntegerList_Type v_DrbIdList:= f_NR_GetActiveDRBs(p_SourcePCellId);
    var template(value) TimingInfo_Type v_TimingInfo;  //@sic R5s201387 sic@
    
    v_Timing := f_NR_GetNextSendOccasion_DL(p_SourcePCellId, 700); // 700 ms // Time for schedule of DL RRC message
    v_Timing1 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 10);  // time for Target SCell-> PCell configuration SS configuration
    v_Timing2 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 20);  // time for Target PCell SS configuration

    // 1. Source PCell: Stop periodic TA.
    // NOTE 1: Unless explicitly specified UL grant configuration keeps configured as per default at the source cell.
    f_NR_ULGrantConfiguration_Common(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, p_UL_GrantConfig_SourceCell);

    // 2. Target PCell: Configure target PCell for no RACH response transmission.
    //Calculate T in the TargetPCell
    v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), p_TargetPCellId);
    f_NR_SS_RachProcedureConfig(p_TargetPCellId, cs_NR_RachProcedureConfig_NoResponse, v_TimingInfo);

    // 3. Source PCell: Schedule the transmission of RRCReconfiguration message to UE requesting Handover to target PCell and SCell.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration, crs_MacBearerRouting(p_SourcePCellId))); //@sic R5-213516 sic@

   // At time T1 if target PCell is the same as source SCell, else at time "Now", steps 4-6:
   if (p_TargetPCellId == p_SourceSCellId) { //@sic R5-223443 R5s220590 sic@
        //Calculate T1 in the TargetPCell
        v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing1), p_TargetPCellId);
        // 4.1 Source PCell: Configure from PCell to normal cell: remove source SCell from the SCell list in the source PCell
        // 5. Target PCell: Configuration of DRBs
        f_NR_ReconfigurePCellToNormalCell( p_SourcePCellId, tsc_PCellIndex_0, v_TimingInfo, v_DrbIdList);
        // 4.2 Target PCell: Configure from SCell to PCell.
        f_NR_ReconfigSCellToNormalCell(p_TargetPCellId, v_RadioBearerList, v_TimingInfo);
   } else {
        // At time Now, steps 5-6:
        // 5. Target PCell: Configuration of DRBs
        f_NR_SS_CommonRadioBearerConfig(p_TargetPCellId, v_RadioBearerList); //@sic R5s201249 sic@
   }

    //6. Transfer of the PDCP Count for AM DRBs and SRBs (if applied) from source to target PCell:
    //   a) Source PCell: Get PDCP COUNT.
    //   b) Target PCell: Set PDCP COUNT.
    //   NOTE 2: No further sending/receiving of DRB data before the HO is done.
    //   NOTE 3: For AM DRBs the PDCP count is maintained, for UM DRBs the PDCP count is reset. For SRBs, the PDCP count is maintained or reset depending on the RRCReconfiguration message content.
    if (p_TargetPCellId == p_SourceSCellId) {
      f_NR_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_RadioBearerList, p_RBConfig_KeyChange, cs_TimingInfo_NR(v_Timing1));
    } else {
      f_NR_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_RadioBearerList, p_RBConfig_KeyChange); //@sic R5s201249 sic@
    }
    
   //At time T2, steps 7-12:
    //7. Target PCell: Inform the SS about the HO and about the source PCell id.
    f_NR_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_NR_PdcpHandoverInit_REQ(p_SourcePCellId, p_TargetPCellId, cs_TimingInfo_NR(v_Timing2)));
   
    //8. Target PCell: Configure RACH procedure either dedicated or C-RNTI based.
    // NOTE 4: The FollowOnFlag is set to true in the ASP reconfiguring C-RNTI
    f_NR_CellInfo_SetRNTI(p_TargetPCellId, p_RNTI_Value);
    //Calculate T2 in the TargetPCell
    v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing2), p_TargetPCellId);
    if (isvalue(p_TargetSCellId)) { //@sic R5s210079 sic@
        //@sic R5s201249 sic@
        f_NR_CellInfo_SetRNTI(valueof(p_TargetSCellId), p_RNTI_Value);
        f_NR_SS_C_RNTI_Config(valueof(p_TargetSCellId), p_RNTI_Value, cs_TimingInfo_NR(v_Timing2)); //@sic R5s210389 sic@
    }
    f_NR_SS_C_RNTI_Config(p_TargetPCellId, p_RNTI_Value, cs_TimingInfo_NR(v_Timing2), tsc_NoCnfReq, tsc_FollowOnFlag);  //@sic R5-206369, R5s210389 sic@
    f_NR_SS_RachProcedureConfigHO(p_TargetPCellId, p_RachProcedureConfig, v_TimingInfo, p_RACH_ConfigDedicated);
    
    //9. Target PCell: Activate security.
    if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s190795 R5s200159 sic@
      f_NR_SS_AS_ActivateSecurity_Current(p_TargetPCellId, p_NCC, v_TimingInfo);
    } else {
      f_NR_SS_AS_ActivateSecurity_NoRefresh(p_TargetPCellId, v_TimingInfo);  //@sic R5s200102 sic@
    }

    //10. Target PCell: Configure UL grant configuration ("OnSR", periodic TA is not started).
    f_NR_ULGrantConfiguration_Common(p_TargetPCellId, v_TimingInfo, -, -, p_UL_GrantConfig_TargetCell);

    //11. Target PCell: Configure Target PCell as PCell.
    //12. Target SCell: Configure Target SCell as SCell with new PCell association.
    //NOTE 5: The FollowOnFlag is set to true in the ASP reconfiguring PCell.
    if (isvalue (p_TargetSCellId)) {  //@sic R5s210079 sic@
      f_NR_ConfigPCell_AddRelSCells(p_TargetPCellId, {valueof(p_TargetSCellId)}, tsc_PCellIndex_0, v_TimingInfo, p_RNTI_Value, -, tsc_NoCnfReq, tsc_FollowOnFlag);  //@sic R5-206369 sic@
      //Calculate T2 in the TargetSCell
      v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing2), valueof(p_TargetSCellId));
      f_NR_ConfigSCell(valueof(p_TargetSCellId), p_SCGConfig_SCell, v_TimingInfo);
    }
    
   // After time T2 (without activation time):
    // 13. Target PCell: Receive RRCReconfigurationComplete.
    // when CA is not active , make sure there is no MAC Bearer Routing ( refer step 7-8 of the test case 8.1.4.1.7.1 )
    if(not isvalue(p_TargetSCellId)) { //@sic R5s220590 sic@
        SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetPCellId, p_RRCConnReconfigComplete));
    } else {
        SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetPCellId, p_RRCConnReconfigComplete, -, crs_MacBearerRouting(p_TargetPCellId))); //@sic R5-213516 sic@
    }
    //14. Target Pell: Start periodic TA.
    f_NR_ULGrantConfiguration_Common(p_TargetPCellId, cs_TimingInfo_Now, -, cs_NR_UplinkTimeAlignment_Start, -);
    
    //15. Target Pell: Inform the SS about completion of the HO (e.g. to trigger PDCP STATUS REPORT PDU).
    f_NR_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_NR_PdcpHandoverComplete_REQ(p_TargetPCellId));

    //16. Target PCell: Re-configure RACH procedure as for initial access.
    f_NR_SS_RachProcedureConfig_Def(p_TargetPCellId);
        
    //17. Source PCell: If source PCell is not target SCell
    //17.1 Reset SRBs and release DRBs.
    if (isvalue (p_TargetSCellId)) {  //@sic R5s210079 sic@
        if (p_SourcePCellId != valueof(p_TargetSCellId)) {
            //17.1  Reset SRBs and release DRBs.
            f_NR_SS_SRBs_DRBs_Release(p_SourcePCellId, cs_TimingInfo_Now, v_DrbIdList);
            f_NR_SS_SRBs_DRBs_Config(p_SourcePCellId, cs_TimingInfo_Now, omit);
        }
    }
    
    //17a. Source PCell:    If target PCell is not the same as source SCell:
    //17a.1 Configure from PCell to normal cell: remove source SCell from the SCell list in the source PCell
    //@sic R5-223443 R5s220590 sic@
    if (p_TargetPCellId != p_SourceSCellId) {
        f_NR_ReconfigurePCellToNormalCell( p_SourcePCellId, tsc_PCellIndex_0, -, v_DrbIdList );
    }
    
    //18. Source SCell: If source SCell is neither target PCell nor target SCell:
    //18.1 Release DRBs
    //@sic R5-223443 R5s220590 sic@
    if (not(match(p_SourceSCellId, p_TargetPCellId)) and not match(p_SourceSCellId, p_TargetSCellId)) {
        f_NR_SS_SRBs_DRBs_Release(p_SourcePCellId, cs_TimingInfo_Now, v_DrbIdList);
    }
   
  }



After Change:
	   function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common(NR_CellId_Type                               p_SourcePCellId,
                                                          NR_CellId_Type                               p_SourceSCellId,
                                                          NR_CellId_Type                               p_TargetPCellId,
                                                          template(omit) NR_CellId_Type                p_TargetSCellId,
                                                          template(omit) CellGroupConfig               p_SCGConfig_SCell,
                                                          RNTI_Value                                   p_RNTI_Value :=  tsc_C_RNTI_Value3,
                                                          boolean                                      p_RBConfig_KeyChange := true,
                                                          template (value) DL_DCCH_Message             p_RRCReconfiguration,
                                                          template (value) NR_RachProcedureConfig_Type p_RachProcedureConfig,
                                                          template (value) NR_DRBInfoList_Type         p_DrbInfoList,
                                                          template (value) UL_GrantConfig_Type         p_UL_GrantConfig_SourceCell := cs_UL_GrantConfig_OnSR,
                                                          template (value) UL_GrantConfig_Type         p_UL_GrantConfig_TargetCell := cs_UL_GrantConfig_OnSR,
                                                          template (present) UL_DCCH_Message           p_RRCConnReconfigComplete := cr_38508_RRCReconfigurationComplete,
                                                          NextHopChainingCount                         p_NCC := tsc_NR_38508_NextHopChainingCount,
                                                          template (omit) ReconfigurationWithSync.rach_ConfigDedicated p_RACH_ConfigDedicated := omit
                                                      ) runs on NR_BASE_PTC
  { //@sic R5s201249: added p_SCGConfig_SCell R5-213516: added MacBearerRouting sic@
    var SubFrameTiming_Type v_Timing;
    var SubFrameTiming_Type v_Timing1;
    var SubFrameTiming_Type v_Timing2;
    var template (value) NR_RadioBearerList_Type v_RadioBearerList := f_NR5GC_GetRadioBearerList(p_DrbInfoList);
    var IntegerList_Type v_DrbIdList:= f_NR_GetActiveDRBs(p_SourcePCellId);
    var template(value) TimingInfo_Type v_TimingInfo;  //@sic R5s201387 sic@
    var NR_PdcpCountInfoList_Type v_PdcpCountInfoList := {}; //WA#WI=973937
    
        
    v_Timing := f_NR_GetNextSendOccasion_DL(p_SourcePCellId, 700); // 700 ms // Time for schedule of DL RRC message
    v_Timing1 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 10);  // time for Target SCell-> PCell configuration SS configuration
    v_Timing2 := f_SubFrameTiming_AddMilliSeconds (v_Timing, 20);  // time for Target PCell SS configuration

    // 1. Source PCell: Stop periodic TA.
    // NOTE 1: Unless explicitly specified UL grant configuration keeps configured as per default at the source cell.
    f_NR_ULGrantConfiguration_Common(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), -, cs_NR_UplinkTimeAlignment_Stop, p_UL_GrantConfig_SourceCell);

    // 2. Target PCell: Configure target PCell for no RACH response transmission.
    //Calculate T in the TargetPCell
    v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), p_TargetPCellId);
    f_NR_SS_RachProcedureConfig(p_TargetPCellId, cs_NR_RachProcedureConfig_NoResponse, v_TimingInfo);

    // 3. Source PCell: Schedule the transmission of RRCReconfiguration message to UE requesting Handover to target PCell and SCell.
    SRB.send(cas_NR_SRB1_RrcPdu_REQ(p_SourcePCellId, cs_TimingInfo_NR(v_Timing), p_RRCReconfiguration, crs_MacBearerRouting(p_SourcePCellId))); //@sic R5-213516 sic@

   // At time T1 if target PCell is the same as source SCell, else at time "Now", steps 4-6:
   if (p_TargetPCellId == p_SourceSCellId) { //@sic R5-223443 R5s220590 sic@
        //Calculate T1 in the TargetPCell
        v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing1), p_TargetPCellId);
        // 4.1 Source PCell: Configure from PCell to normal cell: remove source SCell from the SCell list in the source PCell
        // 5. Target PCell: Configuration of DRBs
        f_NR_ReconfigurePCellToNormalCell( p_SourcePCellId, tsc_PCellIndex_0, v_TimingInfo, v_DrbIdList,true); //WA#WI=973937
        // 4.2 Target PCell: Configure from SCell to PCell.
        f_NR_ReconfigSCellToNormalCell(p_TargetPCellId, v_RadioBearerList, v_TimingInfo);
   } else {
        // At time Now, steps 5-6:
        // 5. Target PCell: Configuration of DRBs
        f_NR_SS_CommonRadioBearerConfig(p_TargetPCellId, v_RadioBearerList); //@sic R5s201249 sic@
   }

    //6. Transfer of the PDCP Count for AM DRBs and SRBs (if applied) from source to target PCell:
    //   a) Source PCell: Get PDCP COUNT.
    //   b) Target PCell: Set PDCP COUNT.
    //   NOTE 2: No further sending/receiving of DRB data before the HO is done.
    //   NOTE 3: For AM DRBs the PDCP count is maintained, for UM DRBs the PDCP count is reset. For SRBs, the PDCP count is maintained or reset depending on the RRCReconfiguration message content.
    if (p_TargetPCellId == p_SourceSCellId) {
      f_NR_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_RadioBearerList, p_RBConfig_KeyChange, cs_TimingInfo_NR(v_Timing1));
    } else {
      f_NR_SS_PdcpCount_Handover(p_SourcePCellId, p_TargetPCellId, v_RadioBearerList, p_RBConfig_KeyChange); //@sic R5s201249 sic@
    }
    
   //At time T2, steps 7-12:
    //7. Target PCell: Inform the SS about the HO and about the source PCell id.
    f_NR_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_NR_PdcpHandoverInit_REQ(p_SourcePCellId, p_TargetPCellId, cs_TimingInfo_NR(v_Timing2)));
   
    //8. Target PCell: Configure RACH procedure either dedicated or C-RNTI based.
    // NOTE 4: The FollowOnFlag is set to true in the ASP reconfiguring C-RNTI
    f_NR_CellInfo_SetRNTI(p_TargetPCellId, p_RNTI_Value);
    //Calculate T2 in the TargetPCell
    v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing2), p_TargetPCellId);
    if (isvalue(p_TargetSCellId)) { //@sic R5s210079 sic@
        //@sic R5s201249 sic@
        f_NR_CellInfo_SetRNTI(valueof(p_TargetSCellId), p_RNTI_Value);
        f_NR_SS_C_RNTI_Config(valueof(p_TargetSCellId), p_RNTI_Value, cs_TimingInfo_NR(v_Timing2)); //@sic R5s210389 sic@
    }
    f_NR_SS_C_RNTI_Config(p_TargetPCellId, p_RNTI_Value, cs_TimingInfo_NR(v_Timing2), tsc_NoCnfReq, tsc_FollowOnFlag);  //@sic R5-206369, R5s210389 sic@
    f_NR_SS_RachProcedureConfigHO(p_TargetPCellId, p_RachProcedureConfig, v_TimingInfo, p_RACH_ConfigDedicated);
    
    //9. Target PCell: Activate security.
    if ((p_NCC > tsc_NR_38508_NextHopChainingCount) or p_RBConfig_KeyChange) { // @sic R5s190795 R5s200159 sic@
      f_NR_SS_AS_ActivateSecurity_Current(p_TargetPCellId, p_NCC, v_TimingInfo);
    } else {
      f_NR_SS_AS_ActivateSecurity_NoRefresh(p_TargetPCellId, v_TimingInfo);  //@sic R5s200102 sic@
    }

    //10. Target PCell: Configure UL grant configuration ("OnSR", periodic TA is not started).
    f_NR_ULGrantConfiguration_Common(p_TargetPCellId, v_TimingInfo, -, -, p_UL_GrantConfig_TargetCell);

    //11. Target PCell: Configure Target PCell as PCell.
    //12. Target SCell: Configure Target SCell as SCell with new PCell association.
    //NOTE 5: The FollowOnFlag is set to true in the ASP reconfiguring PCell.
    if (isvalue (p_TargetSCellId)) {  //@sic R5s210079 sic@
      f_NR_ConfigPCell_AddRelSCells(p_TargetPCellId, {valueof(p_TargetSCellId)}, tsc_PCellIndex_0, v_TimingInfo, p_RNTI_Value, -, tsc_NoCnfReq, tsc_FollowOnFlag);  //@sic R5-206369 sic@
      //Calculate T2 in the TargetSCell
      v_TimingInfo := f_NR_TimingOtherCell(p_SourcePCellId, cs_TimingInfo_NR(v_Timing2), valueof(p_TargetSCellId));
      f_NR_ConfigSCell(valueof(p_TargetSCellId), p_SCGConfig_SCell, v_TimingInfo);
    }
    
   // After time T2 (without activation time):
    // 13. Target PCell: Receive RRCReconfigurationComplete.
    // when CA is not active , make sure there is no MAC Bearer Routing ( refer step 7-8 of the test case 8.1.4.1.7.1 )
    if(not isvalue(p_TargetSCellId)) { //@sic R5s220590 sic@
        SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetPCellId, p_RRCConnReconfigComplete));
    } else {
        SRB.receive(car_NR_SRB1_RrcPdu_IND(p_TargetPCellId, p_RRCConnReconfigComplete, -, crs_MacBearerRouting(p_TargetPCellId))); //@sic R5-213516 sic@
    }
    //14. Target Pell: Start periodic TA.
    f_NR_ULGrantConfiguration_Common(p_TargetPCellId, cs_TimingInfo_Now, -, cs_NR_UplinkTimeAlignment_Start, -);
    
    //15. Target Pell: Inform the SS about completion of the HO (e.g. to trigger PDCP STATUS REPORT PDU).
    f_NR_SS_PdcpHandoverCtrl(p_TargetPCellId, cas_NR_PdcpHandoverComplete_REQ(p_TargetPCellId));

    //16. Target PCell: Re-configure RACH procedure as for initial access.
    f_NR_SS_RachProcedureConfig_Def(p_TargetPCellId);
        
    //17. Source PCell: If source PCell is not target SCell
    //17.1 Reset SRBs and release DRBs.
    if (isvalue (p_TargetSCellId)) {  //@sic R5s210079 sic@
        if (p_SourcePCellId != valueof(p_TargetSCellId)) {
            //17.1  Reset SRBs and release DRBs.
            f_NR_SS_SRBs_DRBs_Release(p_SourcePCellId, cs_TimingInfo_Now, v_DrbIdList);
            f_NR_SS_SRBs_DRBs_Config(p_SourcePCellId, cs_TimingInfo_Now, omit);
        }
    }
    
    //17a. Source PCell:    If target PCell is not the same as source SCell:
    //17a.1 Configure from PCell to normal cell: remove source SCell from the SCell list in the source PCell
    //@sic R5-223443 R5s220590 sic@
    if (p_TargetPCellId != p_SourceSCellId) {
        v_DrbIdList:= f_NR_GetActiveDRBs(p_SourcePCellId); //WA#WI=973937
        f_NR_ReconfigurePCellToNormalCell( p_SourcePCellId, tsc_PCellIndex_0, -, v_DrbIdList, false ); //WA#WI=973937
    }
    
    //18. Source SCell: If source SCell is neither target PCell nor target SCell:
    //18.1 Release DRBs
    //@sic R5-223443 R5s220590 sic@
     if (not(match(p_SourceSCellId, p_TargetPCellId)) and not match(p_SourceSCellId, p_TargetSCellId)) {
        v_PdcpCountInfoList := f_NR_SS_PdcpCount_Get(p_SourceSCellId); //WA#WI=973937
        if(lengthof(v_PdcpCountInfoList) > 0) //WA#WI=973937
        { //WA#WI=973937
            v_DrbIdList:= f_NR_GetActiveDRBs(p_SourceSCellId); //WA#WI=973937
            f_NR_SS_SRBs_DRBs_Release(p_SourceSCellId, cs_TimingInfo_Now, v_DrbIdList); //WA#WI=973937
        }
      } //WA#WI=973937
  }



1.2 [bookmark: _Toc106808479]Correction to the function f_NR_ReconfigurePCellToNormalCell
	Function name
	f_NR_ReconfigurePCellToNormalCell

	Reason for change
	In case of 8.1.4.1.7.1,  during the test steps 4-5, NR Cell 1  NR Cell 3 handover happens with NR Cell 2 added as Scell ; when target Pcell is same as Source Scell , SRB DRB reset is happening twice in the function f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common 
· 1st time SRB DRB RESET inside f_NR_ReconfigurePCellToNormalCell which is called from 4.1,5th steps of f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common
· 2nd  time SRB DRB RESET as part of 17.1 step of f_NR5GC_508RRC_IntraNR_CA_HO_InterCell_Common )  this can be improvised by blocking the redundant RB reset and release inside f_NR_ReconfigurePCellToNormalCell based on the condition

	Summary of change
	SRB DRB reset in the function f_NR_ReconfigurePCellToNormalCell is guarded with the condition so that it avoids the RB RESET on Source Pcell happening twice when source Scell is becoming target Pcell

	TTCN module
	NR_CommonFunctions_CA

	MCC160 Comment
	



Before Change:
	  //------------------------------------------------------------------------------------------------------------------
  /*
   * @desc      Common function to reconfigure a PSCell to a normal cell
   * @param     p_PCell
   * @param     p_PServingCellIndex
   * @param     p_TimingInfo        (default value: cs_TimingInfo_Now)
   * @param     p_DrbToReleassList
   * @status    APPROVED (NR5GC)
   */
   function f_NR_ReconfigurePCellToNormalCell( NR_CellId_Type                          p_PCell,
                                               NR_ServingCellIndex_Type                p_PServingCellIndex,
                                               template (value) TimingInfo_Type        p_TimingInfo := cs_TimingInfo_Now,
                                               IntegerList_Type                        p_DrbToReleassList
                                             ) runs on NR_BASE_PTC
  {
    f_NR_ConfigPCell_AddRelSCells( p_PCell,
                                   {}, //p_SCellList
                                   p_PServingCellIndex,
                                   p_TimingInfo,
                                   omit, // p_RNTI_Value > no need to reconfigure C-RNTI
                                   omit  // p_SS_RadioBearerList > DRB + SRBs to be released just after
                                  );
    //Release DRBs
    f_NR_SS_SRBs_DRBs_Release(p_PCell, p_TimingInfo, p_DrbToReleassList);  //@sic R5s201249 sic@
    //Release and re-establish SRBs - if not ENDC
    if (not f_GetTestcaseAttrib_ENDC(testcasename())) {
        f_NR_SS_SRBs_DRBs_Config(p_PCell, p_TimingInfo, omit);   //@sic R5s201249 sic@
    }
  }



After Change:
	//------------------------------------------------------------------------------------------------------------------
  /*
   * @desc      Common function to reconfigure a PSCell to a normal cell
   * @param     p_PCell
   * @param     p_PServingCellIndex
   * @param     p_TimingInfo        (default value: cs_TimingInfo_Now)
   * @param     p_DrbToReleassList
   * @param     ScellBecomingPcell //WA#WI=973937
   * @status    APPROVED (NR5GC)
   */
   function f_NR_ReconfigurePCellToNormalCell( NR_CellId_Type                          p_PCell,
                                               NR_ServingCellIndex_Type                p_PServingCellIndex,
                                               template (value) TimingInfo_Type        p_TimingInfo := cs_TimingInfo_Now,
                                               IntegerList_Type                        p_DrbToReleassList,
                                               boolean ScellBecomingPcell := false  //WA#WI=973937
                                             ) runs on NR_BASE_PTC
  {
    f_NR_ConfigPCell_AddRelSCells( p_PCell,
                                   {}, //p_SCellList
                                   p_PServingCellIndex,
                                   p_TimingInfo,
                                   omit, // p_RNTI_Value > no need to reconfigure C-RNTI
                                   omit  // p_SS_RadioBearerList > DRB + SRBs to be released just after
                                  );
    //Release DRBs
    if(not ScellBecomingPcell) //WA#WI=973937
    {
        f_NR_SS_SRBs_DRBs_Release(p_PCell, p_TimingInfo, p_DrbToReleassList);  //@sic R5s201249 sic@
        //Release and re-establish SRBs - if not ENDC
        if (not f_GetTestcaseAttrib_ENDC(testcasename())) {
            f_NR_SS_SRBs_DRBs_Config(p_PCell, p_TimingInfo, omit);   //@sic R5s201249 sic@
        }
    } //WA#WI=973937
  }
  









